



Introduction aux calculs astronomiques

et application avec Excel 

- Les coordonnées
- Liens entre les référentiels
- Calcul du temps sidéral
- Position des étoiles
- Position des planètes
- Position de la Lune
- Comparaison avec STELLARIUM
- Pour aller plus loin...
- Annexe



De deux choses lune, l'autre c'est le soleil. Jacques Prévert (Paroles)



Les coordonnées



Pour repérer une étoile en un lieu de la Terre (de *longitude* comptée à partir du méridien de Greenwich et de *latitude* λ) et à une heure donnée¹, on utilise ses coordonnées horizontales :

- la *hauteur* h (par rapport à l'horizon)
- l'*azimut* a (par convention à partir du nord ou du sud)

Pour repérer une étoile sur la voûte céleste indépendamment du lieu et de l'heure, on utilise ses coordonnées équatoriales :

- la *déclinaison* δ (par rapport à l'équateur céleste)
- l'*ascension droite* α mesurée à partir du point vernal γ où se trouve le Soleil à l'équinoxe de printemps²

Dans ce référentiel, l'angle entre le méridien de l'astre et celui du lieu d'observation est appelé l'*angle horaire*³ H



1 – Du fait de la rotation de la Terre autour de son axe, heure et longitude ont le même effet

2 – Une des 2 intersections du plan de l'écliptique avec l'équateur céleste


3 – Lorsque l'angle horaire H est nul, l'étoile est au zénith

Liens entre les référentiels

Les deux référentiels sont liés par le mouvement de la Terre autour de son axe et autour Soleil :

- A une date donnée, on calcule le *temps sidéral de Greenwich* (à 0 h TU¹) puis le *temps sidéral local* tenant compte de l'heure et de la longitude
- Pour chaque objet du ciel, l'angle horaire H s'obtient par la relation :
 $\text{angle horaire } H = \text{temps sidéral local} - \text{ascension droite } \alpha$
- Le passage au référentiel horizontal se fait par rotation d'angle H puis d'angle $90^\circ - \lambda$, d'où :
 $\sin h = \cos \delta \cos H \cos \lambda + \sin \lambda \sin \delta$
 $\sin a = \cos \delta \sin H / \cos h$
 $\cos a = (\cos \delta \cos H \sin \lambda - \cos \lambda \sin \delta) / \cos h$

1 – TU = heure d'hiver à Greenwich



Calcul du temps sidéral

Le calcul se base sur le calendrier julien¹ plus simple pour le calcul du nombre de jours entre 2 dates

Calcul du jour julien de j/m/a :

- Si $m < 3$, $m = m + 12$ et $a = a - 1$
- $A = \text{ENT}^2(a/100)$, $B = 2 - A + \text{ENT}(A/4)$, $C = \text{ENT}(365,25 \times a)$, $D = \text{ENT}(30,6001 \times (m + 1))$
- Jour Julien (JJ) = $B + C + D + j + 1720994,5$

Soit 2451544,5 pour le 1er janvier 2000 à 00h00

Temps sidéral de Greenwich à 0 h TU :

- $T = (JJ - 2451545) / 36525$
- $TS = (24110,54841 + 8640184,812866 \times T + 0,093104 \times T^2 - 0,0000062 \times T^3) / 3600$ (en heures³)
- $TSG = ((TS / 24) - \text{ENT}(TS / 24)) \times 24$ (pour aller de 0 à 24h)

Le temps sidéral local TSL est décalé de l'heure (solaire) et de la longitude :

- $TSL = TSG + \text{Heure} \times 1,0027379 + \text{Longitude}$ (en heures)

1 – En vigueur jusqu'au du 15 octobre 1582, date du passage au calendrier grégorien

2 – ENT = partie entière

3 – et $280,46061837 + 360,98564736629 \times T + 36525 + 0,000387933 \times T^2 - T^3 / 38710000$ en degrés

	A	B	C	D	E	F	G
1	Calcul des positions des étoiles et planètes						
2							
3	Date						
4	24	10	2024				
5	Heure						
6	14	26					
7	Longitude	2	25	=(B7+C7/60)*12/180			
8							
9	Calcul heure sidérale						
10	=SI(B4<3;C4-1;C4)	=SI(B4<3;B4+12;B4)	=A4	1,0027379			
11	A	=ENT(A10/100)					
12	B	=2-B11+ENT(B11/4)					
13	C	=ENT(365,25*A10)					
14	D	=ENT(30,6001*(B10+1))					
15	JJ	=B12+B13+B14+C10+1720994,5					
16	T	=(B15-2451545)/36525					
17	GMST 0hTU	=MOD((24110,54841+(8640184,812866*B16)+(0,093104*(B16^2))-(0,0000062*(B16^3)))/3600;24)/24					
18	TSL	=MOD(HEURE(B17)+MINUTE(B17)/60+(A6+B6/60)*D10+D7;24)					

	A	B	C	D
1	Calcul des positions des étoiles et planètes			
2				
3	Date			
4	24	10	2024	
5	Heure			
6	14	26		
7	Longitude	2	25	0,16
8				
9	Calcul heure sidérale			
10	2024	10	24	1,0027379
11	A	20		
12	B	-13		
13	C	739266		
14	D	336		
15	JJ	2460607,500		
16	T	0,248117728		
17	GMST 0hTU	02:11:34		
18	TSL	16,8172948		

Position des étoiles

- A partir du temps sidéral local, on calcule l'angle horaire de chaque étoile connaissant son ascension droite et sa déclinaison
- Avec la latitude, on calcule sa hauteur et son azimut
- Dans XCIEL.xls, pour les vues de type carte, on projette orthogonalement sur le plan de l'équateur les points de la sphère céleste (projection orthographique). Cette projection déforme les constellations lorsque l'on s'approche de l'équateur



	A	B	C	D	E	F	G	H	I	J	K	L
1	TSL	16.817295										
2												
3	Latitude	48	50	0.85								
4												
5	Etoiles et constellations											
6												
7	Nom	AsH	AsM	DeD	DeM	AsC	DeC	AnglHor	Haut	SinAzim	CosAzim	Azim
8	Achernar	1	36	-57	-5	=+B8+C8/60	=(D8+E8/60)*PI()/180	=SI(\$B\$1-F8<0;24+\$B\$1-F8;\$B\$1-F8)	=ASIN(COS(G8)*COS(H8*PI()/12)*COS(\$D\$3)+SIN(\$D\$3)*SIN(G8))	=(COS(G8)*SIN(H8*PI()/12))/COS(\$D\$3)	=(COS(G8)*COS(H8*PI()/12)*SIN(\$D\$3)-ASIN(J8)+PI()/2)/COS(\$D\$3)	=SI(K8<0;-ASIN(J8)+PI()/2;ASIN(J8))
9	Acrux	12	23	-62	-8	=+B9+C9/60	=(D9+E9/60)*PI()/180	=SI(\$B\$1-F9<0;24+\$B\$1-F9;\$B\$1-F9)	=ASIN(COS(G9)*COS(H9*PI()/12)*COS(\$D\$3)+SIN(\$D\$3)*SIN(G9))	=(COS(G9)*SIN(H9*PI()/12))/COS(\$D\$3)	=(COS(G9)*COS(H9*PI()/12)*SIN(\$D\$3)-ASIN(J9)+PI()/2)/COS(\$D\$3)	=SI(K9<0;-ASIN(J9)+PI()/2;ASIN(J9))



	A	B	C	D	E	F	G	H	I	J	K	L
1	TSL	16.817295										
2												
3	Latitude	48	50	0.85								
4												
5	Etoiles et constellations											
6												
7	Nom	AsH	AsM	DeD	DeM	AsC	DeC	AnglHor	Haut	SinAzim	CosAzim	Azim
8	Achernar	1	36	-57	-5	1,60	-1,00	15,22	-1,06	-0,82	0,57	-0,97
9	Acrux	12	23	-62	-8	12,38	-1,08	4,43	-0,57	0,51	0,86	0,54
10	Adara	6	56	-28	-9	6,93	-0,49	9,88	-1,01	0,88	-0,48	2,07
11	Agena	14	0	-60	-1	14,00	-1,05	2,82	-0,42	0,37	0,93	0,38
12	Aldebaran	4	33	16	4	4,55	0,28	12,27	-0,44	-0,07	-1,00	3,22

Position des planètes

- Dans XCIEL.xls, les calculs sont simplifiés¹

Le calcul de la position des planètes se fait à partir de leurs positions héliocentriques à une date donnée² et en tenant compte de leurs données orbitales et de la date.

Pour l'ascension droite, on calcule l'*élongation* (angle apparent par rapport au Soleil)

$$\tan \text{Elongation} = \frac{\sin(\text{PosHélio P} - \text{PosHélio T})}{\frac{D_{\text{Sol T}}}{D_{\text{Sol P}}} - \cos(\text{PosHélio P} - \text{PosHélio T})}$$

La déclinaison s'obtient en tenant compte de l'angle de l'axe de rotation de la Terre avec l'écliptique (23° 26')


On obtient alors les coordonnées équatoriales à partir de celles du Soleil et on est ramené au calcul précédent

1 – Les trajectoires des planètes sont considérées circulaires et coplanaires (approximation valide jusqu'à Neptune)

2 – Voir [Ephémérides du Soleil, de la Lune et des Planètes - Xavier Jubier](https://ssp.imcce.fr/forms/ephemeris) et <https://ssp.imcce.fr/forms/ephemeris>



	A	B	C	D	E	F	G	H
1	Date	01/01/2000						
2								
3	Planètes	AsC Soleil Greenwich	01/01/2000 00:00	18h42m54sec		soit	=18+42/60+54/3600	
4								
5	Nom	DSol (UA)	Période (j)	Teta0	PosHélio	Elong	AsC	DeC
6	Soleil						=MOD((PI()+E8)*12/PI();24)	=ASIN(SIN(G6*PI()/12)*SIN((23+26/60)*PI()/180))
7	Vénus	0,72	224,70	-187	=MOD(D7+(((\$B\$1-\$E\$3)/C7-ENT((\$B\$1-\$E\$3)/C7))*360;360)*PI()/180	=SI((\$B\$8/B7)>(COS(E7-\$E\$8)):ATAN(SIN(E7-\$E\$8)/(((\$B\$8/B7)-COS(E7-\$E\$8)))):ATAN(SIN(E7-\$E\$8)/(((\$B\$8/B7)-COS(E7-\$E\$8)))+PI()	=MOD(\$G\$6-F7*12/PI();24)	=ASIN(SIN(G7*PI()/12)*SIN((23+26/60)*PI()/180))
8	Terre	1,00	365,26	101	=MOD(D8+(((\$B\$1-\$E\$3)/C8-ENT((\$B\$1-\$E\$3)/C8))*360;360)*PI()/180			
9	Mars	1,52	686,98	0	=MOD(D9+(((\$B\$1-\$E\$3)/C9-ENT((\$B\$1-\$E\$3)/C9))*360;360)*PI()/180	=SI((\$B\$8/B9)>(COS(E9-\$E\$8)):ATAN(SIN(E9-\$E\$8)/(((\$B\$8/B9)-COS(E9-\$E\$8)))):ATAN(SIN(E9-\$E\$8)/(((\$B\$8/B9)-COS(E9-\$E\$8)))+PI()	=MOD(\$G\$6-F9*12/PI();24)	=ASIN(SIN(G9*PI()/12)*SIN((23+26/60)*PI()/180))



	A	B	C	D	E	F	G	H
1	Date	01/01/2000						
2								
3	Planètes	AsC Soleil Greenwich	01/01/2000 00:00	18h42m54s		soit		18,715
4								
5	Nom	DSol (UA)	Période (j)	Teta0	PosHélio	Elong	AsC	DeC
6	Soleil						18,715	-0,40
7	Vénus	0,72	224,70	-187	3,02	0,72	15,951	-0,35
8	Terre	1,00	365,26	101	1,76			
9	Mars	1,52	686,98	0	0,00	-0,86	22,01	-0,20

Position de la Lune (1/3)

Le calcul de la trajectoire de la Lune est, en raison de la proximité du Soleil, un problème à 3 corps qui n'admet pas de solution analytique



Les valeurs caractéristiques de l'orbite de la Lune sont :

- sa période de révolution 27,32166 jours
- son demi-grand axe de 384 748 km
- son inclinaison de $5,145^\circ$ sur l'écliptique
- la période de régression de la ligne des nœuds de 18,6 ans
- son excentricité moyenne de 0,0549

Du fait des perturbations dues au Soleil, ces valeurs sont variables au cours du temps

Les éclipses de Soleil et de Lune ne peuvent se produire que lorsque les nœuds sont alignés (ou au plus près) avec l'axe Terre-Soleil

Repérage de l'orbite d'un satellite



Position de la Lune (2/3)

Le calcul est simplifié en considérant circulaire l'orbite de la Lune :

On calcule la position orbitale de la Lune au temps t par :
 $l = t \times 360^\circ / \text{Période} + l_0 - \Omega$ avec $\Omega = \Omega_0 + \dot{\Omega} \times t$ où $\dot{\Omega} = -360^\circ / 18,613 / 365,25$

Le vecteur Terre-Lune de coordonnées $\begin{pmatrix} \cos l \\ \sin l \\ 0 \end{pmatrix}$ dans le plan orbital de la Lune devient dans le plan de l'écliptique


$$\begin{pmatrix} x = \cos \Omega \cos l - \sin \Omega \sin l \cos i \\ y = \sin \Omega \cos l + \cos \Omega \sin l \cos i \\ z = \sin l \sin i \end{pmatrix}$$

par rotation d'angle i autour de x puis d'angle Ω autour de z

On passe au plan équatorial terrestre par rotation d'angle ε

autour de x'
$$\begin{pmatrix} x' = x \\ y' = y \cos \varepsilon - z \sin \varepsilon \\ z' = y \sin \varepsilon + z \cos \varepsilon \end{pmatrix} = \begin{pmatrix} \cos \delta \cos \alpha \\ \cos \delta \sin \alpha \\ \sin \delta \end{pmatrix}$$

Avec les coordonnées équatoriales, déclinaison δ et ascension droite α , on est ramené au calcul précédent



Les coordonnées sphériques sont l'ascension droite α et la déclinaison δ quand le plan de référence est l'équateur et la longitude λ et la latitude β quand le plan de référence est l'écliptique

Position de la Lune (3/3)

Formules :




A	B	C	D	E	F	G	H	I	J	K	L	M
Date	Heure	Référence	Période (j)	Inclinaison (°)	Ω (°)	$\dot{\Omega}$ (°/jour)	I_0 (°)	Inclinaison Terre (°)	Année (j)			
19/06/2006	00:00	01/01/2000 12:00:00	27,32166	5,145	124,9955279	=-360/18,613/J2	43,05258785	=23+26/60	=365+6/24+9/(24*60)+10/(24*3600)			
Nb jours	Ω (°)	I (°)	$\cos I$	$\sin I$	x	y	z	x'	y'	z'	AsC	DeC
=A2+B2-C2	=MOD(A5*\$G\$2+\$F\$2,360)	=A5*360/\$D\$2-B5*\$H\$2	=COS(C5*PI()/180)	=SIN(C5*PI()/180)	=COS(B5*PI()/180)*D5-SIN(B5*PI()/180)*E5	=SIN(B5*PI()/180)*D5+COS(B5*PI()/180)*E5	=E5*SIN(-\$E\$2*PI()/180)	=F5	=G5*COS(-\$I\$2*PI()/180)-H5*SIN(-\$I\$2*PI()/180)	=G5*SIN(-\$I\$2*PI()/180)+H5*COS(-\$I\$2*PI()/180)	=(ATAN2(I5;J5)+PI())*12/PI()	=ASIN(K5/(I5^2+J5^2+K5^2)^0,5)



A	B	C	D	E	F	G	H	I	J	K	L	M
Date	Heure	Référence	Période (j)	Inclinaison (°)	Ω (°)	$\dot{\Omega}$ (°/jour)	I_0 (°)	Inclinaison	Année (j)			
19/06/2006	00:00	01/01/2000 12:00	27,32166	5,145	124,9955	-0,053	43,05258785	23,4333	365,26			
Nb jours	Ω (°)	I (°)	$\cos I$	$\sin I$	x	y	z	x'	y'	z'	AsC	DeC
2360,50	0,001	31145,847	-0,995	-0,102	-0,995	-0,101	0,009	-0,995	-0,089	0,049	0,343	0,049

Pendant le cycle lunaire (lunaison), la déclinaison de la Lune varie. Elle augmente pendant une moitié du cycle et décroît pendant l'autre moitié


L'inclinaison des plans orbitaux conduit à deux extrêmes d'environ $\pm 28,5^\circ$ et $\pm 18,5^\circ$ appelés lunistiques majeur (plus grande hauteur dans l'hémisphère nord) et mineur




- A — Axe de rotation de la Terre
- B — Plan orbital de la Lune autour de la Terre
- C — Plan orbital de la Terre autour du Soleil (plan de l'écliptique)
- D — Plan équatorial de la Terre

Comparaison avec STELLARIUM




Carte du ciel le 20 10 2024 à 22 h 0 m TU



Carte du ciel le 20 10 2024 à 16 h 0 m TU



Carte du ciel le 20 10 2024 à 13 h 0 m TU





Pour aller plus loin...

- Pdf téléchargeables :

- [Introduction aux codes astronomiques](#) de Michel Llibre
- [Mécanique Spatiale](#) (CNES Ecole d'été de Mathématiques 2012)

- Ouvrages de référence :

- « Calculs astronomiques à l'usage des amateurs » par Jean Meeus de la Société astronomique de France
- [PyMeeus](#) : librairie Python des calculs de Jean Meeus
- « Astronomie méthodes et calculs » de Agnès Acker et Carlos Jaschek de l'université de Strasbourg



Annexe

- **Un peu de trigonométrie...**

- Un angle α est repéré sur un cercle de rayon 1 par deux coordonnées $\sin \alpha$ et $\cos \alpha$
- On a en radians :
 $\sin(0) = 0$ et $\cos(0) = 1$
 $\sin(\pi/2) = 1$ et $\cos(\pi/2) = 0$
- On a toujours : $\cos^2(\alpha) + \sin^2(\alpha) = 1$ (Pythagore)
 $\sin(-\alpha) = -\sin(\alpha)$ et $\cos(-\alpha) = \cos(\alpha)$

- **Pour faire une rotation de repère dans le plan x, y ...**

Les coordonnées du point M deviennent :

$$x' = x \cos \theta + y \sin \theta$$

$$y' = -x \sin \theta + y \cos \theta$$

$$z' = z$$

En 3D, on combine les rotations

